

Example 5: Crossed Random Effects Models for Trials nested within Subjects and within Items (complete data, syntax, and output available for SAS, SPSS, and STATA electronically)

Source: Locker Jr., L., Hoffman, L., & Bovaird, J. A. (2007). On the use of multilevel modeling in the analysis of psycholinguistic data. *Behavior Research Methods*, 39(4), 723-730.

Response time data for a lexical decision task (decide as quickly as you can whether this is a word or a non-word) were collected for 39 items from 38 subjects (total possible observations = 1482; total actual observations = 1392 after removing inaccurate responses). Items are words that varied systematically in two characteristics: Semantic Frequency (low/high) and Neighborhood Size (small/large).

SAS Data Manipulation:

```
* SAS: Bringing in data from folder to work library;
* Adding another version of predictors to be coded 0/1 for low/high;
DATA example5; SET example.Example5;
    IF freq=-.5 THEN freq01=0; IF freq=.5 THEN freq01=1;
    IF size=-.5 THEN size01=0; IF size=.5 THEN size01=1;
RUN;
```

SPSS Data Manipulation:

```
* SPSS code to import data.
GET FILE = "example/Example5.sav".
DATASET NAME example5 WINDOW=FRONT.

* Adding another version of predictors to be coded 0/1 for low/high.
RECODE freq size (-.5=0) (.5=1) INTO freq01 size01.
EXECUTE.
```

STATA Data Manipulation:

```
* Adding another version of predictors to be coded 0/1 for low/high
gen freq01 = freq
gen size01 = size
recode freq01 (-.5=0)
recode freq01 ( .5=1)
recode size01 (-.5=0)
recode size01 ( .5=1)
```

Model 1: Empty means baseline model with only residual variance $\rightarrow RT_{tis} = \gamma_{000} + e_{tis}$
(default REPEATED statement if not included is TYPE=VC)

```
TITLE1 "SAS Empty Means Model: No Random Intercepts (E only)";
PROC MIXED DATA=example5 COVTEST NOCLPRINT NOITPRINT NAMELEN=100 IC METHOD=REML;
    CLASS Subject Item;
    MODEL rt = / SOLUTION DDFM=Satterthwaite;
    ODS OUTPUT InfoCrit=FitOnly; * Save fit for comparison;
RUN; TITLE1;
```

```
TITLE "SPSS Empty Means Model: No Random Intercepts (E only)".
MIXED rt BY Subject Item
    /METHOD = REML
    /PRINT = SOLUTION TESTCOV
    /FIXED = .
```

```
* STATA Empty Means Model: No Random Intercepts (E only)
xtmixed rt , variance reml,
    estimates store Eonly // save for LRT
```

SAS Output:

Covariance Parameter Estimates					
Cov Parm	Estimate	Standard Error	Z Value	Pr > Z	
Residual	21340	809.19	26.37	<.0001	All the variance in RT in one pile of e (TYPE=VC)

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
17820.7	1	17822.7	17822.7	17824.7	17827.9	17828.9

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	632.38	3.9154	1391	161.51	<.0001

Model 2: Is there significant mean RT variation across subjects? → $RT_{tis} = \gamma_{000} + U_{00s} + e_{tis}$

```

TITLE1 "SAS Random Intercept for Subjects Model";
PROC MIXED DATA=example5 COVTEST NOCLPRINT NOITPRINT NAMELEN=100 IC METHOD=REML;
  CLASS Subject Item;
  MODEL rt = / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT /SUBJECT=Subject TYPE=UN; * Level 2 variance for subjects;
  ODS OUTPUT InfoCrit=FitRandSubjects; * Save fit for comparison;
RUN; TITLE1;

TITLE "SPSS Random Intercept for Subjects Model".
MIXED rt BY Subject Item
  /METHOD = REML
  /PRINT = SOLUTION TESTCOV
  /FIXED =
  /RANDOM = INTERCEPT | SUBJECT(Subject) COVTYPE(UN).

* STATA Random Intercept for Subjects Model
xtmixed rt , || subject: , variance reml,
  estimates store RandSubjects, // save for LRT
  lrtest RandSubjects Eonly // Test random subjects

```

SAS Output:

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	subject	5167.07	1305.09	3.96	<.0001
Residual		16307	626.74	26.02	<.0001

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
17540.3	2	17544.3	17544.3	17545.4	17547.5	17549.5

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	631.42	12.1540	37	51.95	<.0001

If total RT variance = 21,474, then

5,167 / 21,474 = 24% is between subjects

16,307 / 21,474 = 76% is within subjects

Is there significant variation in mean RT across subjects—is that new 24% > 0%?

```
* Calculate difference in model fit relative to e-only model;
%FitTest(FitFewer=FitEonly, FitMore=FitRandSubjects);
```

Likelihood Ratio Test for FitEonly vs. FitRandSubjects

	Neg2Log						
Name	Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitEonly	17820.7	1	17822.7	17827.9	.	.	.
FitRandSubjects	17540.3	2	17544.3	17547.5	280.439	1	0

Note that in this case, this LRT for the improvement in model fit appears elsewhere on the page!

Model 3: Is there significant mean RT variation across items? $\rightarrow RT_{tis} = \gamma_{000} + U_{00s} + U_{0i0} + e_{tis}$

```
TITLE1 "SAS Random Intercepts for Subjects and Items: Crossed Model";
PROC MIXED DATA=example5 COVTEST NOCLPRINT NOITPRINT NAMELEN=100 IC METHOD=REML;
  CLASS Subject Item;
  MODEL rt = / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / SUBJECT=Item TYPE=UN;          * Level 2 variance for items;
  RANDOM INTERCEPT / SUBJECT=Subject TYPE=UN;        * Level 2 variance for subjects;
  ODS OUTPUT InfoCrit=FitRandItems CovParms=CovEmpty; * Save fit, variances to compare;
RUN; TITLE1;
```

```
TITLE "SPSS Random Intercepts for Subjects and Items: Crossed Model".
```

```
MIXED rt BY Subject Item
  /METHOD = REML
  /PRINT = SOLUTION TESTCOV
  /FIXED =
  /RANDOM = INTERCEPT | SUBJECT(Subject) COVTYPE(UN)
  /RANDOM = INTERCEPT | SUBJECT(Item) COVTYPE(UN).
```

```
* STATA Random Intercept for Subjects and Items: Crossed Model
xtmixed rt , || _all: R.subject, ||_all: R.item, variance reml,
  estimates store RandItems,          // save for LRT
  lrtest RandItems RandSubjects      // Test random items
```

SAS Output:

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	item	2409.36	678.04	3.55	0.0002
UN(1,1)	subject	5166.81	1292.78	4.00	<.0001
Residual		14344	559.99	25.61	<.0001

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
17439.9	3	17445.9	17445.9	17439.9	17439.9	17442.9

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	635.33	14.4301	59.4	44.03	<.0001

If total variance now = 21,920, then...

5,167 / 21,920 = 24% is between subjects

2,409 / 21,920 = 11% is between items

14,344 / 21,920 = 65% is within subjects and items (subject x item interaction)

Is there significant variation in mean RT across items—is that new 11% > 0%?

```
* Calculate difference in model fit relative to random subjects model;
%FitTest(FitFewer=FitRandSubjects, FitMore=FitRandItems);
```

Likelihood Ratio Test for FitRandSubjects vs. FitRandItems

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitRandSubjects	17540.3	2	17544.3	17547.5	.	.	.
FitRandItems	17439.9	3	17445.9	17439.9	100.399	1	0

Calculate 95% random effects confidence intervals for each random intercept:

95% CI = fixed effect \pm 1.96*SQRT(variance)

Subject Intercept CI = $635 \pm 1.96 \cdot \text{SQRT}(5167) = 494$ to 776

95% of the individual subject mean RTs are expected to fall between 494 and 776 ms

Item Intercept CI = $635 \pm 1.96 \cdot \text{SQRT}(2409) = 539$ to 732

95% of the individual item mean RTs are expected to fall between 539 and 732 ms

Model 4a: Are there significant fixed effects of the item predictors Frequency and Size?

$\rightarrow RT_{tis} = \gamma_{000} + \gamma_{010}(\text{Freq}_i) + \gamma_{020}(\text{Size}_i) + \gamma_{030}(\text{Freq}_i)(\text{Size}_i) + U_{00s} + U_{0i0} + e_{tis}$

```
TITLE1 "SAS Random Subjects by Random Items Crossed Predictive Model";
TITLE2 "Using CLASS statement to get cell means and comparisons VIA LSMEANS";
PROC MIXED DATA=example5 COVTEST NOCLPRINT NOITPRINT NAMELEN=100 IC METHOD=REML;
  * Add freq and size to CLASS statement to use LSMEANS;
  CLASS Subject Item freq01 size01;
  * | operator estimates all possible main effects and interactions up to @ order;
  MODEL rt = freq01|size01@2 / SOLUTION DDFM=Satterthwaite OUTPM=ItemPred;
  RANDOM INTERCEPT / SUBJECT=Item TYPE=UN;      * Level 2 variance for items;
  RANDOM INTERCEPT / SUBJECT=Subject TYPE=UN;   * Level 2 variance for subjects;
  ODS OUTPUT InfoCrit=FitRandItem CovParms=CovItemPred; * Save fit, variances to compare;
  * Requesting marginal means per condition (what Type 3 tests are for);
  LSMEANS freq01 size01;
  * Requesting F-tests for simple main effects (more useful than marginal);
  LSMEANS freq01*size01 / SLICE=freq01 SLICE=size01;
RUN; TITLE1; TITLE2;
```

TITLE "SPSS Random Intercepts for Subjects and Items Crossed Predictive Model".

SUBTITLE "Using BY statement to get cell means and comparisons via EMMEANS".

```
MIXED rt BY Subject Item freq01 size01
  /METHOD = REML
  /PRINT = SOLUTION TESTCOV
  /FIXED = freq01 size01 freq01*size01
  /RANDOM = INTERCEPT | SUBJECT(Subject) COVTYPE(UN)
  /RANDOM = INTERCEPT | SUBJECT(Item) COVTYPE(UN)
  /EMMEANS = TABLES(freq01)
  /EMMEANS = TABLES(size01)
  /EMMEANS = TABLES(freq01*size01) COMPARE(freq01)
  /EMMEANS = TABLES(freq01*size01) COMPARE(size01).
```

* STATA Random Intercepts for Subjects and Items Crossed Predictive Model

```
xtmixed rt ib(last).freq01 ib(last).size01 i.freq01#i.size01, ///
  || _all: R.subject, ||_all: R.item, variance reml,
  estimates store NewRandItem, // save for LRT
  margins i.freq01 i.size01, // marginal means
  contrast i.freq01 i.size01, // marginal mean diff tests
  margins i.freq01#i.size01, // cell means
  pwcompare i.freq01#i.size01, pv // cell mean diff tests
```

SAS Output:

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z	
UN(1,1)	item	1692.07	526.60	3.21	0.0007	Intercept Variance across ITEMS in mean RT
UN(1,1)	subject	5168.48	1293.11	4.00	<.0001	Intercept Variance across SUBJECTS in mean RT
Residual		14341	559.79	25.62	<.0001	Leftover trial-to-trial residual variance

Information Criteria


Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
17402.4	3	17408.4	17408.5	17402.4	17402.4	17405.4

Solution for Fixed Effects → are SIMPLE MAIN EFFECTS (highest=0 given CLASS statement)

Effect	freq01	size01	Estimate	Standard Error	DF	t Value	Pr > t
Intercept			618.21	19.1504	58.8	32.28	<.0001
freq01	0		2.0097	20.9460	31.2	0.10	0.9242
freq01	1		0
size01		0	67.5951	21.1176	32.2	3.20	0.0031
size01		1	0
freq01*size01	0	0	-72.0301	29.3756	31.8	-2.45	0.0199
freq01*size01	0	1	0
freq01*size01	1	0	0
freq01*size01	1	1	0

Type 3 Tests of Fixed Effects → THESE MAIN EFFECTS ARE NOW MARGINAL

Effect	Num DF	Den DF	F Value	Pr > F
freq01	1	31.8	5.36	0.0272
size01	1	31.8	4.62	0.0393
freq01*size01	1	31.8	6.01	0.0199

**Least Squares Means → Means per condition and/or cell as requested**

Effect	freq01	size01	Estimate	Standard Error	DF	t Value	Pr > t	
freq01	0		618.00	15.5006	62.5	39.87	<.0001	MARGINAL MEANS
freq01	1		652.01	15.7322	63.5	41.44	<.0001	
size01		0	650.79	15.5588	63.4	41.83	<.0001	
size01		1	619.21	15.6749	62.7	39.50	<.0001	
freq01*size01	0	0	615.78	18.5739	60.7	33.15	<.0001	CELL MEANS
freq01*size01	0	1	620.22	18.5482	60.3	33.44	<.0001	
freq01*size01	1	0	685.80	18.7416	62.7	36.59	<.0001	
freq01*size01	1	1	618.21	19.1504	58.8	32.28	<.0001	

Tests of Effect Slices → TESTS OF SIMPLE MAIN EFFECTS

Effect	freq01	size01	Num DF	Den DF	F Value	Pr > F	
freq01*size01	0		1	31.4	0.05	0.8295	size effect for low freq
freq01*size01	1		1	32.2	10.25	0.0031	size effect for high freq
freq01*size01		0	1	32.4	11.56	0.0018	freq effect for small size
freq01*size01		1	1	31.2	0.01	0.9242	freq effect for large size

* Get total R²;

PROC CORR DATA=ItemPred; VAR pred rt; RUN;

rt	Pred	rt	
	0.17421	1.00000	→ .17421 ² = Overall R ² = .03
Response Time in Milliseconds	<.0001		

```
* Calculate PseudoR2 relative to empty means model;
%PseudoR2(NCov=3, CovFewer=CovEmpty, CovMore=CovItemPred);
```

PseudoR2 (% Reduction) for CovEmpty vs. CovItemPred

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
CovEmpty	UN(1,1)	item	2409.36	678.04	3.55	0.0002	.
CovEmpty	UN(1,1)	subject	5166.81	1292.78	4.00	<.0001	.
CovEmpty	Residual		14344	559.99	25.61	<.0001	.
CovItemPred	UN(1,1)	item	1692.07	526.60	3.21	0.0007	0.29771
CovItemPred	UN(1,1)	subject	5168.48	1293.11	4.00	<.0001	-0.00032
CovItemPred	Residual		14341	559.79	25.62	<.0001	0.00018

Why didn't we explain any subject or residual variance?

Model 5: Should items still be treated as a random effect?

Is there still significant variance in mean RT across items after controlling for frequency and size?

$$\rightarrow RT_{tis} = \gamma_{000} + \gamma_{010}(\text{Freq}_i) + \gamma_{020}(\text{Size}_i) + \gamma_{030}(\text{Freq}_i)(\text{Size}_i) + U_{00s} + e_{tis}$$

```
TITLE1 "SAS Dropping Random Item Intercept";
PROC MIXED DATA=example5 COVTEST NOCLPRINT NOITPRINT NAMELEN=100 IC METHOD=REML;
  CLASS Subject Item;
  MODEL rt = freq01|size01@2 / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / SUBJECT=Subject TYPE=UN; * Level 2 variance for subjects ONLY;
  ODS OUTPUT InfoCrit=FitNoRandItem; * Save fit to compare;
RUN; TITLE1;
```

```
TITLE "SPSS Dropping Random Item Intercept".
MIXED rt BY Subject Item freq01 size01
  /METHOD = REML
  /PRINT = SOLUTION TESTCOV
  /FIXED = freq01 size01 freq01*size01
  /RANDOM = INTERCEPT | SUBJECT(SUBJECT) COVTYPE(UN).
```

```
* STATA Dropping Random Item Intercept
xtmixed rt ib(last).freq01 ib(last).size01 i.freq01#i.size01, ///
  || subject: , variance reml,
  estimates store NoRandItem, // save for LRT
  lrtest NewRandItem NoRandItem // Test random items
  margins i.freq01 i.size01, // marginal means
  contrast i.freq01 i.size01, // marginal mean diff tests
  margins i.freq01#i.size01, // cell means
  pwcompare i.freq01#i.size01, pv // cell mean diff tests
```

SAS Output:

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	subject	5171.97	1302.28	3.97	<.0001
Residual		15688	603.61	25.99	<.0001

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
17466.4	2	17470.4	17470.4	17471.6	17473.7	17475.7

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	614.64	13.3976	54.8	45.88	<.0001
freq01	62.5713	9.5910	1352	6.52	<.0001
size01	5.4273	9.2634	1351	0.59	0.5580
freq01*size01	-64.6343	13.4592	1351	-4.80	<.0001

Type 3 Tests of Fixed Effects				
Effect	Num DF	Den DF	F Value	Pr > F
freq01	1	1351	20.22	<.0001
size01	1	1351	15.97	<.0001
freq*size	1	1351	23.06	<.0001

Is there still significant item variance remaining?

```
* Calculate difference in model fit relative to random subjects and items predictive model;
%FitTest(FitFewer=FitNoRandItem, FitMore=FitRandItem);
```

Likelihood Ratio Test for FitNoRandItem vs. FitRandItem

Name	Neg2Log Likelihood	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitNoRandItem	17466.4	2	17470.4	17473.7	.	.	.
FitRandItem	17402.4	3	17408.4	17402.4	63.9914	1	1.2212E-15

Model 6: Is there a significant random subject slope for the item predictor of frequency?

```
TITLE1 "SAS Random Slope for Effect of Freq over Subjects";
TITLE2 "Removed Freq and Size from CLASS to fit random slope";
PROC MIXED DATA=example5 COVTEST NOCLPRINT NOITPRINT NAMELEN=100 IC METHOD=REML;
  CLASS Subject Item;
  MODEL rt = freq01|size01@2 / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / SUBJECT=Item TYPE=UN;          * Level 2 variance for items;
  RANDOM INTERCEPT freq01 / SUBJECT=Subject TYPE=UN; * Level 2 variances for subjects;
  ODS OUTPUT InfoCrit=FitRandFreq;                    * Save fit to compare;
RUN; TITLE1; TITLE2;
```

```
TITLE "SPSS Random Slope for Effect of Freq over Subjects".
SUBTITLE "Removed Freq and Size from BY statement to fit random slope".
MIXED rt BY Subject Item WITH freq01 size01
  /METHOD = REML
  /PRINT = SOLUTION TESTCOV
  /FIXED = freq01 size01 freq01*size01
  /RANDOM = INTERCEPT freq01 | SUBJECT(Subject) COVTYPE(UN)
  /RANDOM = INTERCEPT | SUBJECT(Item) COVTYPE(UN).
```

* This code should have worked (I think), but it didn't converge

```
* STATA Random Slope for Effect of Freq over Subjects
* Removed Freq and Size from i. statement to fit random slope
xtmixed rt c.freq01 c.size01 c.freq01#c.size01, ///
  || subject: freq01, || item: , variance reml,
  estimates store RandFreq, // save for LRT
  lrtest RandFreq NewRandItem // Test random freq
```

SAS Output:

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	item	1700.03	527.91	3.22	0.0006
UN(1,1)	subject	5231.22	1307.42	4.00	<.0001
UN(2,1)	subject	1058.11	571.78	1.85	0.0642
UN(2,2)	subject	371.65	447.45	0.83	0.2031
Residual		14244	563.58	25.28	<.0001

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
17397.6	5	17407.6	17407.6	17397.6	17397.6	17402.6

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	615.85	17.9378	55.8	34.33	<.0001
freq01	69.8447	20.8577	33.5	3.35	0.0020
size01	4.4434	20.4461	31.4	0.22	0.8294
freq01*size01	-72.0683	29.4136	31.8	-2.45	0.0200

Does the effect of frequency vary over subjects?

* Calculate difference in model fit relative to random subjects and items predictive model;
 %FitTest(FitFewer=FitRandItem, FitMore=FitRandFreq);

Likelihood Ratio Test for FitRandItem vs. FitRandSize

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitRandItem	17402.4	3	17408.4	17402.4	.	.	.
FitRandSize	17402.4	5	17412.4	17402.4	0.081671	2	0.95999

Calculate 95% random effects confidence intervals for the frequency effect across subjects:

95% CI = fixed effect \pm 1.96*SQRT(variance)

Subject Frequency Slope CI = $69.84 \pm 1.96*\text{SQRT}(371.65) = 32 \text{ to } 107$

95% of the individual subject simple frequency slopes are expected to fall between 32 and 107 ms

Writing out a single-level combined equation for this last model to illustrate the random slopes:

$$RT_{tis} = \gamma_{000} + \gamma_{010}(\text{Freq}_i) + \gamma_{020}(\text{Size}_i) + \gamma_{030}(\text{Freq}_i)(\text{Size}_i) + U_{00s} + U_{01s}(\text{Freq}_i) + U_{0i0} + e_{tis}$$

$$RT_{tis} = 615.85 + (69.84*\text{Freq}_i) + (4.44*\text{Size}_i) + (-72.07*\text{Freq}_i*\text{Size}_i)$$

+ U_{00s} → increment to *mean RT* depending on which subject after controlling for NOTHING
 + U_{0i0} → increment to *mean RT* depending on which item after controlling for freq and size
 + $U_{01s}(\text{Freq}_i)$ → increment to *slope of frequency* depending on which subject
 + e_{tis} → increment to *trial RT* depending on which trial after controlling for everything

Sample Results Section:

The extent to which semantic frequency (coded low = 0, high = 1) and phonological neighborhood size (coded small = 0, large = 1) could predict response time (RT) in milliseconds in a lexical decision task was examined for 39 items administered to 38 subjects. Because RTs for incorrect responses were not included, the data were unbalanced, such that each subject had a different number of trials included for each condition. Accordingly, rather than aggregating the individual trial RTs into potentially biased item condition means (that would assume items are fixed) and conducting an analysis of variance, all possible RTs were examined instead in a multilevel model with crossed random effects, in which individual trials (the combination of each subject with each item) were nested within subjects and within items, which were crossed random effects. Restricted maximum likelihood within SAS PROC MIXED was used to estimate all models; denominator degrees of freedom were estimated with the Satterthwaite method.

The extent to which systematic variability in mean RT existed for each dimension of sampling was first examined in a series of empty means models (i.e., only a fixed intercept and no predictors). Relative to a model with only a residual variance, the addition of a random intercept variance for subjects significantly improved model fit, $-2\Delta LL(\sim 1) = 280.4$, $p < .001$ (and the smaller AIC and BIC concur), indicating significant differences between subjects in mean RT, and that trials from the same subject were positively correlated. The addition of a random intercept for items also significantly improved model fit, $-2\Delta LL(\sim 1) = 100.4$, $p < .001$ (and the smaller AIC and BIC concur), indicating significant differences between items in mean RT as well, and that trials for the same item were also positively correlated. Of the total estimated RT variance, 24% was due to between-subject differences in mean RT (given by the subject random intercept), 11% was due to between-item differences in mean RT (given by the item random intercept), and the remaining 65% was due to the subject by item interaction (i.e., residual variance). Construction of 95% random effects confidence intervals as described in Snijders and Bosker (1999) revealed that 95% of subject mean RTs are expected to fall between 494 and 776 ms, whereas 95% of the item mean RTs are expected to fall between 539 and 732 ms. Thus, there was relatively more variability across subjects than across items. The extent to which the main and interaction effects of semantic frequency and neighborhood size could account for between-item differences in mean RT was then examined in a conditional model; results are provided in Table 1.

ANOVA-like description of the results: There was a significant semantic frequency by neighborhood size interaction, $F(1,31.8) = 6.01$, $p = .0199$; the pattern of the interaction is shown in Figure 1 and was decomposed by examining simple main effects of each predictor. First, with respect to the effect of neighborhood size, for low frequency words, there was no significant difference between words with small or large neighborhood size ($M = 615.78$, $M = 620.22$), $F(1,31.4) = 0.05$, $p = .8295$, whereas for high frequency words, responses were significantly slower to words with smaller than larger neighborhoods ($M = 685.80$, $M = 618.21$), $F(1,32.2) = 10.25$, $p = .0031$. With respect to the effect of frequency, for small neighborhood words, responses were significantly faster to words of low than high frequency ($M = 615.78$, $M = 685.80$), $F(1,32.4) = 11.56$, $p = .0018$, whereas for large neighborhood words, there was no significant difference between words of low or high frequency ($M = 620.22$, $M = 618.21$), $F(1,31.2) = 0.01$, $p = .9242$.

Regression-like description of the same results: The fixed intercept for the predicted RT for a word of low frequency and small size was $\gamma_{000} = 615.78$. There was a significant simple main effect for the mean difference between low and high frequency words of small size of $\gamma_{010} = 70.02$ ($p = .002$). There was a nonsignificant simple main effect for the mean difference between small and large size words of low frequency of $\gamma_{020} = 4.44$ ($p = .830$). However, there was a significant frequency by size interaction of $\gamma_{030} = -72.03$ ($p = .020$), such that relative to the frequency effect for small words of $\gamma_{010} = 70.02$, the frequency effect for large words was significantly less positive by -72.03 (yielding a nonsignificant simple effect of frequency for large words of $\gamma_{010} + \gamma_{030} = -2.01$, $p = .924$). Similarly, relative to the size effect for low frequency words of $\gamma_{020} = 4.44$, the size effect for high frequency words was significantly more negative by -72.03 (yielding a significant simple effect of size for high frequency words of $\gamma_{020} + \gamma_{030} = -67.56$, $p = .003$). Thus, as shown in Figure 1, a positive frequency effect was found only for words of small size, and a negative size effect was found only for high frequency words.

The effects of frequency and size explained approximately 30% of the item intercept variance. Given that 11% of the total RT variance was due to mean differences between items, this translates into a total reduction in all RT variance of 3.28%. The extent to which these effects were sufficient to describe all between-item differences in mean RT was then examined by removing the item random intercept variance from the conditional model. The resulting significant decrease in model fit, $-2\Delta LL(\sim 1) = 64.4$, $p < .001$ (and the larger AIC and BIC) suggest that significant differences remain between items after controlling for their primary design features, or that items should not be treated as fixed effects. Finally, the potential for individual subject differences in the frequency effect was examined by adding a random subject frequency slope (and its covariance with the subject random intercept) to the model. Model fit did not significantly improve, $-2\Delta LL(\sim 2) = 4.8$, $p = .091$ (although the AIC and BIC were smaller), indicating that each subject does not need his or her own random deviation from the fixed effect of frequency.